

Altair SLC Python procedure user guide

Version: 2023.5

Copyright 2002-2023 World Programming, an Altair Company

www.altair.com

Contents

- PYTHON procedure..... 3**
 - Introduction..... 3
 - Setup and configuration..... 4
 - Using Python with Altair SLC..... 5
 - Data type conversion..... 6
 - Import custom Python modules..... 8
 - Using graphics created by Python..... 10
 - Example..... 10
 - Python procedure reference..... 12
 - PROC PYTHON..... 13
 - EXECUTE..... 14
 - EXPORT..... 16
 - IMPORT..... 17
 - SUBMIT..... 18
 - ENDSUBMIT..... 18

- Legal Notices..... 20**

PYTHON procedure

The PYTHON procedure enables SAS language programs to include code written in the Python language.

Combining the Python language and the SAS language enables the bulk of a data processing and analytics solution to be written in the industrial strength and high-performing SAS language, while also exploiting features present in the Python language.

Introduction ↗	3
The Python procedure enables SAS language programs to include code written in the Python language.	
Setup and configuration ↗	4
Setting the Python environment for Altair SLC.	
Using Python with Altair SLC ↗	5
Using Python in a SAS language program enables you to make use of specialist Python packages such as <code>Scikit-learn</code> or <code>Tensorflow</code> .	
Example ↗	10
Demonstrates how to use a SAS language dataset in <code>PROC PYTHON</code> to create a scatter plot diagram.	
Python procedure reference ↗	12
Describes the syntax and options for <code>PROC PYTHON</code> and its contained statements.	

Introduction

The Python procedure enables SAS language programs to include code written in the Python language.

By combining Python and the SAS language you can:

- Use the SAS language to bulk process and prepare data, and pass the processed data to Python.
- Use Python packages you have previously developed for data analysis.
- Use Python data analysis packages or solutions that may not be available in the SAS language.

Data is passed between the SAS language and Python language environments using the `EXPORT` statement. Once data has been transferred, that data is made available as a pandas DataFrame to a Python program. On completion of the Python program data can, if required, be returned to the SAS language environment using the `IMPORT` statement.

Setup and configuration

Setting the Python environment for Altair SLC.

When using Python with Altair SLC:

- The `pandas` package must be installed with the Python interpreter. This can be checked using the `pip` utility by running `pip list` on the command line.

The Altair SLC distribution does not include either the Python interpreter or `pandas` package. If you do not have Python installed, you can obtain a copy of the Python interpreter from <https://www.python.org> or install a packaged Python environment that includes the necessary modules.

The `PYTHON` procedure can be used with Python version 3.5.0 and later, and is currently supported on Microsoft Windows, Linux-based systems, and macOS.

The procedure is not currently supported on IBM mainframe.

Python environment variables

You must set the `PYTHONHOME` environment variable for Altair SLC to locate and use the Python interpreter. This variable must reference the folder where the main Python library is located – for example, `python3.dll` on Microsoft Windows.

Standard output and error streams

The Python standard output stream (`sys.stdout`) and standard error stream (`sys.stderr`) are redirected to Altair SLC output when the procedure is run:

- `sys.stderr` is redirected to the log file.
- `sys.stdout` is redirected to the listing file when Altair SLC is run on the command line, or to all specified ODS output destinations when Workbench is used.

If you use the Python `print()` function to put output to the listing file, the number of characters appearing in the listing output is determined by the `LINESIZE` system option. The `LINESIZE` option can be used to print strings of up to 256 characters; the use of the `print()` function should therefore be for limited information such as log statements.

If you want to return large volumes of Python-generated content to Altair SLC, create a `DataFrame` containing the required content and import the `DataFrame` using the `IMPORT` statement.

The following example creates a list of functions available in the Python pandas package. The list is printed using `print (fnList)` and also converted to a DataFrame.

```
PROC PYTHON;
SUBMIT;
import inspect
fnPandas = inspect.getmembers(pandas, inspect.isfunction)
fnList = [fn[0] for fn in fnPandas]
print (fnList)
fnTable = pandas.DataFrame({'function': fnList})
ENDSUBMIT;
IMPORT DATA=PANDAFN PYTHON=fnTable;
RUN;
```

The printed output truncates the list of functions:

```
['Expr', 'Term', 'bdate_range', 'concat', 'crosstab',
 'cut', 'date_range', 'eval', 'factorize',
```

When converted to a DataFrame and imported, all available functions are listed.

Using Python with Altair SLC

Using Python in a SAS language program enables you to make use of specialist Python packages such as `Scikit-learn` or `Tensorflow`.

The first time the Python procedure is invoked in a SAS language program, the `pandas` and `numpy` packages are automatically imported. You can access the functionality in the `pandas` and `numpy` packages in an in-line Python language program – written between the `SUBMIT` and `ENDSUBMIT` statements – by referencing the fully-qualified package, class or function name, for example:

```
PROC PYTHON;
  SUBMIT;
content = pandas.read_csv('file:///C:/project/sourcedata/example.csv')
...
  ENDSUBMIT;
RUN;
```

Alternatively, you can use the `import ... as ...` statement to alias either the `pandas` or `numpy` package name, for example:

```
PROC PYTHON;
  SUBMIT;
import pandas as pd
content = pd.read_csv('file:///C:/project/sourcedata/example.csv')
...
  ENDSUBMIT;
RUN;
```

Other Python packages can be imported and used within the in-line Python code, for example:

```
PROC PYTHON;
  EXPORT DATA=source;
  SUBMIT;
import statsmodels.formula.api as lm
result = lm.ols(formula='x ~ y + z', data=source).fit()
...
  ENDSUBMIT;
RUN;
```

Each subsequent use of the PYTHON procedure in a SAS language program can use the same Python environment. This means any global variables or imported packages used in a PYTHON procedure invocation are available to all subsequent PYTHON procedure invocations.

Each PYTHON procedure invocation can include multiple blocks of in-line Python language code, and use a combination of in-line Python language code, and Python programs run using the EXECUTE statement.

Data type conversion

Describes the correspondence between SAS language formats and pandas data types.

This section describes the correspondence between SAS language formats and Python pandas data types. Altair SLC has many formats that affect the output and display of data. When you write data to a pandas DataFrame using the Python procedure EXPORT statement, formatted data is converted to an equivalent pandas or numpy data type.

Many formats only affect the layout of data output, such as adding currency symbols or comma separators, and these formats have no effect when writing data.

SAS language unformatted data to Python

Unformatted data is converted to a pandas DataFrame type as follows:

SAS language format	Python data type	Notes
Unformatted number	float64	
Unformatted string	object	The maximum object length for a variable is not known as part of the dataset metadata.

SAS language formatted data to Python – numbers

The core numeric formats are converted to a pandas DataFrame type as follows:

SAS language format	Python data type	Notes
<i>w. d</i>	Float64	
BEST. and BEST <i>w.</i>	float64	
FLOAT <i>w. d</i>	float64	

SAS language formatted data to Python – strings

The core character formats are converted to a pandas DataFrame type as follows:

SAS language format	Python data type	Notes
<i>\$w. \$CHARw. \$Fw.</i>	object	The maximum object length for a variable is not known as part of the dataset metadata.

SAS language formatted data to Python – dates and times

Date, datetime and time formats are converted to a pandas DataFrame type as follows:

SAS language format	Python data type	Notes
DATE <i>w.</i>	datetime64 [ns]	Numpy datetime type.
DDMMYY <i>w.</i> and all variants (such as DDMMYYB <i>w.</i> , MMDDYY <i>Sw.</i> , and YYMM <i>w.</i>)	datetime64 [ns]	Numpy datetime type.
DTDATE <i>w.</i> and all variants (such as DTMONYY <i>w.</i> and DTWKDATX <i>w.</i>	datetime64 [ns]	Numpy datetime type.
TIME <i>w.</i> , HOUR <i>w.</i> , HHMM <i>w.</i> and all similar time formats.	float64	
JULIAN <i>w.</i> and all similar date formats.	datetime64 [ns]	Numpy datetime type.

Python pandas data types to SAS language dataset

SAS language datasets only contain numeric and character data. Formats might be applied to the data in the dataset to more closely represent the source data from Python. Importing a pandas DataFrame using the `IMPORT` statement converts data as follows:

Pandas data type	SAS language format	Notes
Object	Character	The maximum object length is calculated before importing.
int64	Numeric	
bool	Numeric	<i>True</i> is converted to 1; <i>False</i> is converted to zero (0).

Pandas data type	SAS language format	Notes
Float64	Numeric	
datetime64[ns]	Numeric	Formatted as DATETIME19.

Some values in Python cannot be represented in the same manner as a SAS language datasets. The following table shows how these values are converted:

Pandas data value	SAS language value	Notes
null string (' ')	.	Missing character value.
True	1	Numeric value
False	0 (zero)	Numeric value
NaN Defined with <code>float('nan')</code>	.	Missing numeric value
Infinity Defined with <code>float('inf')</code> Defined with <code>float('-inf')</code>	.	Missing numeric value

Import custom Python modules

How to use your own packages and modules in Python.

To use custom packages, they must be accessible to Python. The paths to locations containing custom packages are specified by the Python `sys.path` variable. The `sys.path` variable constructs a list of locations to search using the `PYTHONHOME` and `PYTHONPATH` environment variables, or the locations can be specified by modifying the value of the variable during the execution of a program.

- `sys.path` contains a list of folders searched for packages. The variable is constructed from the values in `PYTHONHOME` and `PYTHONPATH`.
- `PYTHONHOME` specifies the location of the Python interpreter and standard libraries, including packages installed into *site-packages* using a package manager such as *PIP*. The variable must be specified for Altair SLC to interact with Python.
- `PYTHONPATH` specifies a list of folders to search for packages. This list is prepended to the search list defined in `sys.path`.

The first item in `sys.path` is either the directory containing the python program or an empty string, interpreted by Python as the current directory. When run from Altair SLC, `sys.path[0]` contains the first search path specified in either `PYTHONPATH` or `PYTHONHOME`. If your program references packages in the current directory, you must modify `sys.path` when `PROC PYTHON` is running. See [Modifying the `sys.path` variable](#) (page 9)

Set *PYTHONPATH* before running Altair SLC

The *PYTHONPATH* variable can be defined as a system variable, and used by all applications on your device that run Python.

If you have multiple installations of Python, setting *PYTHONPATH* using a system variable might cause your program to attempt to import the incorrect version of a package. In these circumstances, you should set the variable as part of the SAS language program you run in Altair SLC.

Set *PYTHONPATH* in a SAS language program

The *PYTHONPATH* can be set in a SAS language program using the `SET` system option. For example, to use packages stored in `C:\temp\python` folder, the following can be added to a program before the `PROC PYTHON` statement:

```
OPTIONS SET = PYTHONPATH 'C:\temp\python';
```

If *PYTHONHOME* is specified as `C:\python3`, the content of the Python `sys.path` variable is:

```
['C:\\temp\\python', 'C:\\python3\\python37.zip',  
'C:\\python3\\DLLs', 'C:\\python3\\lib', 'C:\\python3',  
'C:\\python3\\lib\\site-packages']
```

Modifying the *sys.path* variable

The Python *sys.path* variable can be modified programmatically by adding the following to a Python language program:

```
import os, sys  
sys.path.append(os.getcwd())
```

If a Python language program run using the `EXECUTE` statement includes other Python files, add the above as an in-line program before the executed program to enable Altair SLC to locate the included files:

```
PROC PYTHON;  
  SUBMIT;  
import os, sys  
sys.path.append(os.getcwd())  
  ENDSUBMIT;  
  EXECUTE 'programs/dataCollect.py';  
RUN;
```

Using graphics created by Python

Any graphics generated using functionality in Python are captured and can be included in ODS output.

When the procedure runs it creates the temporary variable *wpsgloc* that points to a temporary folder used to store graphics for inclusion in ODS output. The variable must be added to the name of the image file every time you create an image. This can be done using, for example, the `os.path.join()` Python function:

```
PROC PYTHON;
SUBMIT;
import os
import matplotlib.pyplot as plt
...
plt.savefig(os.path.join(wpsgloc, 'my_image.png'))
ENDSUBMIT;
RUN;
```

The *wpsgloc* contains one of the following:

- The WORK location if the procedure is run in Altair Analytics Workbench; for example:

```
C:\Users\user-id\AppData\Local\Temp\Data
```

- The working directory if a program containing the procedure is run on the command line.

The variable cannot be modified in the SAS language program.

In the following program, the images created by the PYTHON procedure are written to the PDF file specified in the ODS statement.

```
ODS PDF FILE='scatter_plot.pdf';
PROC PYTHON;
  EXPORT DATA=STATS PYTHON=df;
  SUBMIT;
import os
import matplotlib.pyplot as plt
df.plot(kind='scatter', x='i', y='j')
plt.savefig(os.path.join(wpsgloc, 'scatter.png'))
  ENDSUBMIT;
RUN;
ODS PDF CLOSE;
```

Example

Demonstrates how to use a SAS language dataset in PROC PYTHON to create a scatter plot diagram.

The following example creates a dataset in a SAS language DATA step, and then uses the EXPORT statement to pass that dataset to the Python environment. The dataset is converted to a pandas DataFrame as part of the export, and the DataFrame is used to create a scatter plot using Matplotlib.

An output PDF file destination is created using the SAS language Output Delivery System (ODS). Adding PDF to the output destinations includes the returned scatter plot image file in the PDF output. The PDF is saved and the output can be viewed in a PDF viewer.


This example requires the following Python packages:

- SciPy
- Matplotlib

```
ODS PDF FILE='scatter_plot.pdf';
DATA stats (drop = count);
  DO count=1 TO 10;
    DO numCount=1 TO 10;
      numVal = (numCount*numCount)+(count*count);
      OUTPUT;
    END;
  END;
RUN;

PROC PYTHON;
  EXPORT DATA=STATS PYTHON=df;
  SUBMIT;
import os
import matplotlib.pyplot as plt
df.plot(kind='scatter', x='numCount', y='numVal')
plt.savefig(os.path.join(wpsgloc, 'scatter.png'))
  ENDSUBMIT;
RUN;
ODS PDF CLOSE;
```

This creates the following scatter plot in the ODS PDF output:



Python procedure reference


Describes the syntax and options for PROC PYTHON and its contained statements.

PROC PYTHON ↗	13
Invokes the Python environment that enables the execution of in-line or external Python language programs.	
EXECUTE ↗	14
Runs a Python program stored in a separate file.	
EXPORT ↗	16
Converts a SAS language dataset to a pandas DataFrame.	
IMPORT ↗	17
Enables a pandas DataFrame to be converted to a SAS language dataset and referenced in a SAS language program.	

SUBMIT ↗	18
Specifies the start of an in-line Python language program.	
ENDSUBMIT ↗	18
Specifies the end of an in-line Python language program.	

PROC PYTHON

Invokes the Python environment that enables the execution of in-line or external Python language programs.



Datasets created in Altair SLC can be made available to the Python program using the `EXPORT` statement, and a dataset imported from the Python program into Altair SLC using the `IMPORT` statement.

A Python program can be either written in-line in the `PYTHON` procedure, or run from a separate file:

- To run an in-line Python program, use the `SUBMIT` and `ENDSUBMIT` statements.
- To run a Python program stored in an external file use the `EXECUTE` statement.

The Python environment is exited using a `RUN` statement.

When the Python environment is invoked, the `pandas` and `numpy` packages are automatically loaded. These packages can be used either by quoting the full package name in code, or by using the `import ... as ...` statement to alias the package name.

Options

The following options are available with the `PROC PYTHON` statement.

KEEP

Specifies that the current Python environment is not terminated when the procedure exits.



When specified, the current Python environment is kept active when the current procedure exits, and the environment is used in the next invocation of the Python procedure in the same program. If that invocation does not specify `KEEP`, the environment is terminated when the procedure exits.

The default behaviour is to terminate the Python environment at the end of the procedure. Specifying `KEEP` keeps the current Python environment, including any modules loaded during the execution of a Python program, to be used in the next invocation of the `PYTHON` procedure.

You can specify the `PYTHONKEEP` system option to use the same Python environment for the duration of the execution of the SAS language program.

LIB


Specifies the library location for datasets used in the PYTHON procedure. The default location is the WORK library.



The *library-reference* location is used when *libname* is not specified as part of the path for the DATA option of either the EXPORT or IMPORT statements.

TERMINATE

Specifies that the Python environment is terminated when the procedure exits.



Specifying TERMINATE stops the current Python environment even if the PYTHONKEEP system option has been specified. A subsequent invocation of PROC PYTHON in the same program only has the default pandas and numpy packages loaded.


Example

The following example shows how to invoke the PYTHON procedure to print hello world to ODS output.

```
PROC PYTHON;
SUBMIT;
print ('Hello World')
ENDSUBMIT;
RUN;
```

EXECUTE

Runs a Python program stored in a separate file.



The EXECUTE statement is an alternative to using the SUBMIT statement. It enables Python code placed in a separate file to be run Altair SLC.

Execute options

The following options are available with the EXECUTE statement.

filename

A string, in quotation marks, containing the path of the Python program file. *filename* can be either an absolute pathname or a relative pathname.

In Workbench, the path for relative file pathnames is the Workspace. For example, to run a file named `myProgram.py` from a project named `python`, the relative path is `/python/myProgram.py`.

cmd-argument

Specifies a command line argument passed to the Python program. Arguments are accessed in the programing from `sys.argv`.

All command line arguments are passed to the Python program as strings. Numeric arguments must be in quotation marks, and the Python program must convert the arguments to the required numeric type.

Basic example

In this example, a Python program stored in an external file is executed in the `PYTHON` procedure. The file is referenced using the absolute path:

```
PROC PYTHON;  
  EXECUTE 'C:\temp\printDS.py';  
RUN;
```

Example – pass an argument to a Python program

In this example, an external Python program `multiple.py` is executed that returns the square of a specified value. The value is specified in a variable.

```
import sys  
  
def multiply (value):  
  return value*value  
  
print(multiply(int(sys.argv[0])))
```

`multiple.py` is executed from the `PYTHON` procedure in a SAS language program, and the required value to multiply is passed to the Python program as an argument to the `EXECUTE` statement. The numeric argument `6` is passed as a string. The Python program uses the built-in `int()` function to return an integer before calculating the square.


```
PROC PYTHON;  
  EXECUTE 'C:/temp/multiple.py' '6';  
RUN;
```

This produces the following in ODS output:

```
36
```

EXPORT

Converts a SAS language dataset to a pandas `DataFrame`.



Dataset preparation should be completed before exporting the data to Python. This enables you to use the dataset processing capability of the SAS language to create an export dataset containing only the required data for the Python language program.

Export options

The following options are available with the `EXPORT` statement.

DATA

Specifies the dataset location and name of the SAS language dataset to be converted.

The library location can be specified using either `libname` in the `DATA` option, or the `LIB` option of the `PROC PYTHON` statement.

- If `libname` is specified, that location is always used.
- If the `LIB` option of the `PROC PYTHON` statement is specified and `libname` is not specified, the location in the `LIB` option is used.
- If neither `libname` nor the `LIB` option on the `PROC PYTHON` statement are specified, the `WORK` location is used.

PYTHON

Specifies the name of the pandas `DataFrame` as used in the Python language program.

The pandas `DataFrame` name in Python language code is case sensitive, and the `dataframe-name` specified must match the case used of the Python variable name.

If this option is not specified, the `dataframe-name` default is the `dataset` name specified in the `DATA` option. If you use the default `dataset` name in an in-line Python language program, the variable name must match the case used in the `DATA` option.

Example – export a SAS language dataset to a pandas DataFrame

The following example creates a dataset in a SAS language DATA step. The dataset is then exported to a pandas DataFrame and the column types printed out.

```
DATA TESTDATA;
INFILE CARDS DLM='#';
INPUT num char $;
CARDS;
1 # Hello
2 # World
;
RUN;


PROC PYTHON;
EXPORT DATA=TESTDATA PYTHON=dfname;
SUBMIT;
print (dfname)
ENDSUBMIT;
RUN;
```

Which writes the following to ODS output:

```
   num      char
0  1.0      Hello
1  2.0      World
```

IMPORT

Enables a pandas DataFrame to be converted to a SAS language dataset and referenced in a SAS language program.



Import options

The following options are available with the `IMPORT` statement.

DATA

Specifies the dataset location and name as used in Altair SLC.

The library location can be specified using either *libname* in the `DATA` option, or the `LIB` option of the `PROC PYTHON` statement.

- If *libname* is specified, that location is always used.

- If the `LIB` option of the `PROC PYTHON` statement is specified and `libname` is not specified, the location in the `LIB` option is used.
- If neither `libname` nor the `LIB` option on the `PROC PYTHON` statement are specified, the `WORK` location is used.


PYTHON

Specifies the name of the pandas `DataFrame` as used in the Python environment. Must be specified.

`dataframe-name` is case sensitive and must match the case used for the imported pandas `DataFrame` in the Python program.

SUBMIT

Specifies the start of an in-line Python language program.



An in-line program is defined as part of the `PYTHON` procedure in a SAS language program. The `SUBMIT` statement marks the start of the program, and the `ENDSUBMIT` statement marks the end.


The Python language program must start on a new line after the `SUBMIT` statement, and the `ENDSUBMIT` statement must appear at the beginning of a line on its own. The first line of the Python program code must not start with any white space and any subsequent statements must follow the normal Python requirements for indentation, for example:

```
PROC PYTHON;
  SUBMIT;
  fruits = ['apple', 'banana', 'cherry', 'damson', 'elderberry', 'fig']
  for fruit in fruits:
    print(fruit)
  ENDSUBMIT;
RUN;
```

Multiple in-line Python language programs can exist in a single `PYTHON` procedure. Each Python language program is executed as it is encountered. Variables defined in one in-line language program can be used in subsequent in-line programs in the same Python procedure.

ENDSUBMIT

Specifies the end of an in-line Python language program.



The `ENDSUBMIT` statement must be entered at the start of a new line after the Python language program.

Legal Notices

Copyright 2002-2023 World Programming, an Altair Company

This information is confidential and subject to copyright. No part of this publication may be reproduced or transmitted in any form or by any means, electronic or mechanical, including photocopying, recording, or by any information storage and retrieval system.

Trademarks

Altair SLC™, Altair Analytics Workbench™, and Altair SLC Hub™ are registered trademarks or trademarks of Altair Engineering, inc. (r) or ® indicates a Community trademark.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

All other trademarks are the property of their respective owner.

General Notices

Altair Engineering, inc. is not associated in any way with the SAS Institute.

Altair SLC is not the SAS System.

The phrases "SAS", "SAS language", and "language of SAS" used in this document are used to refer to the computer programming language often referred to in any of these ways.

The phrases "program", "SAS program", and "SAS language program" used in this document are used to refer to programs written in the SAS language. These may also be referred to as "scripts", "SAS scripts", or "SAS language scripts".

The phrases "IML", "IML language", "IML syntax", "Interactive Matrix Language", and "language of IML" used in this document are used to refer to the computer programming language often referred to in any of these ways.

Altair SLC includes software developed by third parties. More information can be found in the THANKS or acknowledgments.txt file included in the Altair SLC installation.